

	
3GPP TSG-RAN5 Meeting #2022-TTCN email	R5s220609
Online, , 13th Dec 2021 - 31st Dec 2022
	CR-Form-v12.2

	CHANGE REQUEST

	

	
	34.229-3
	CR
	0827
	rev
	-
	Current version:
	16.5.0
	

	

	For HELP on using this form: comprehensive instructions can be found at 
http://www.3gpp.org/Change-Requests.

	



	Proposed change affects:
	UICC apps
	
	ME
	
	Radio Access Network
	
	Core Network
	



	

	Title:	
	Addition of IMS 5GS emergency call  test case 10.9

	
	

	Source to WG:
	ANRITSU LTD

	Source to TSG:
	R5

	
	

	Work item code:
	TEI15_Test, 5GS_NR_LTE-UEConTest
	
	Date:
	2022-04-26

	
	
	
	
	

	Category:
	B 
	
	Release:
	Rel-16

	
	Use one of the following categories:
F  (correction)
A  (mirror corresponding to a change in an earlier 													release)
B  (addition of feature), 
C  (functional modification of feature)
D  (editorial modification)
Detailed explanations of the above categories can
be found in 3GPP TR 21.900.
	Use one of the following releases:
Rel-8	(Release 8)
Rel-9	(Release 9)
Rel-10	(Release 10)
Rel-11	(Release 11)
…
Rel-16	(Release 16)
Rel-17	(Release 17)
Rel-18	(Release 18)
Rel-19	(Release 19)

	
	

	Reason for change:
	To add Emergency Call test case 10.9.

	
	

	Summary of change:
	This document lists all changes applied to Emergency Call test case 10.9 required for approval. See detailed change description for further information

	
	

	Consequences if not approved:
	Test case will not be added to ATS

	
	

	Clauses affected:
	10.9.NR5GC

	
	

	
	Y
	N
	
	

	Other specs
	
	X
	 Other core specifications	
	TS/TR ... CR ... 

	affected:
	
	X
	 Test specifications
	TS/TR ... CR ...

	(show related CRs)
	
	X
	 O&M Specifications
	TS/TR ... CR ... 

	
	

	Other comments:
	

	
	

	This CR's revision history:
	



Page 1


[bookmark: _Toc100832109]Table of Contents
Table of Contents	2
1	Overview	3
1.1	Verification Test Summary	3
2	Corrections required	3
2.1	Change 1	3
2.2	Change 2	6
2.3	Change 3	7
3	Branches executed	8
4	Execution Log Files	9
4.1	Qualcomm SM 8450+SDX65 5G UE	9
5	References	9


1. [bookmark: _Toc122434485][bookmark: _Toc100832110]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 10.9 which is part of the NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
[bookmark: _Hlk90298041]Contact:	Kalahasti, Narendra
	Narendra.Kalahasti@anritsu.com
0. [bookmark: _Toc100832111]Verification Test Summary 
Test Case: 	10.9
ATS Version:	iwd-TTCN3-B2020-09_D22wk12
System Simulator used:	Anritsu Protocol Conformance Test System ME7834NR,
UE used:	Qualcomm SM 8450+SDX65 5G UE
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc100832112]Corrections required
1. [bookmark: _Toc30685521][bookmark: _Toc97137453][bookmark: _Toc100832113]Change 1
	Function name
	f_TC_10_9_NR5GC_IMS2

	Reason for change
	
1. Resource reservation, PDU Session Modification, will happen during call stablishment. With current implementation it may happen that PDU Session Modification procedure has not finished before Call is released (BYE is sent from SS).
2. After the call is released it is necessary to inform IPCAN and also to handle coordination trigger sent by IPCAN – both done by f_IMS_CC_EndCall.
3. Another coordination trigger to IPCAN is needed to coordinate PDU session release.

	Summary of change
	1. Added a delay of 5s to give some time for PDU Session Modification Complete to NR PTC.
2. Added f_IMS_CC_EndCall in f_TC_10_9_NR5GC_IMS2 after the call is released.
3. Added coordination trigger to IPCAN

	TTCN module
	IMS\NR5GC_10\IMS_NR5GC_EmergencyCallTestcases.ttcn

	MCC160 Comment
	Not needed.  Alternative solution in R5s220531 



Before Change
	  
    function f_TC_10_9_NR5GC_IMS2() runs on IMS_PTC
  { /* IMS implementation for PDN2 - Emergency call */
    var SipUrl v_ContactUrlUE;
    var template (value) MessageHeader v_MessageHeader;
    var template (value) MessageBody v_MessageBody;
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var REGISTER_Request v_RegisterRequest;
    var boolean v_CondA7 := true;
    var boolean v_GeolocationUsage := pc_IMS_Geolocation; /* the test case intends to use GeoLocationInformation if supported by the UE (PICS) @sic R5s150846 sic@
                                                             INVITE (step 1 of Annex C.22) refers to condition A8 of A.2.1 */

    f_IMS_PTC_Init(PDN_2);
    
    // Test Case Body
    f_IMS_TestBody_Set(true);
    
    f_IMS_IMS_WaitForTrigger(OtherPDN);
    // @siclog "Steps 2-9" siclog@
    // Steps 1-8 of Table 4.9.11.2.2-1 of TS 38.508-1 are performed.
    // EXCEPTION: In parallel to the events described in steps 9-10 below the events specified in step of Table 4.9.11.2.2-2 of TS 38.508-1 [21] take place
    // @siclog "Step 11-12" siclog @
    // Steps 9-10 of Table 4.9.11.2.2-1 of TS 38.508-1 are performed.
    //EXCEPTION: In parallel to the events described in steps 12-20  below the events specified in 11-13 of annex Table 4.9.11.2.2-1 of TS 38.508-1[21] are performed.
    f_IMS_CC_StartCall(IPCAN_EmergencyCall_NoRegistration); // 4.9.11.2.2-2 of TS 38.508-1
    
    //@siclog "Step 12" siclog@
    //UE sends REGISTER(Step 1 of Annex A.3)
    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest(-, omit, -, -, -, v_CondA7);
    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;
    
    // Reconfig of UE client ports in case the UE uses UDP and different than default (5060) ports
    f_IMS_Register_SecurityInit(v_RegisterRequest);
    
    //@siclog "Step 13" siclog@
    //SS sends 401 Unauthorized(Step 2 of Annex A.3)
    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterRequest))));

    //@siclog "Step 14" siclog@
    //UE sends REGISTER(Step 3 of Annex A.3)
    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest(-, -, v_CondA7);
    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;
    
    //@siclog "Step 15" siclog@
    //SS sends 403 Forbidden
    v_MessageHeader := f_IMS_RegisterResponse_403_MessageHeaderTX(v_RegisterRequest);
    v_MessageHeader.contentType := cs_ContentType(c_ims3gppApplication);
    v_MessageBody := cs_MessageBody_XML_Ims_3gpp(cs_Ims_3gpp_Emergency("anonymous-emergencycall"));
    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine403, v_MessageHeader, v_MessageBody)));
    
    f_IMS_GlobalRelease (oldestProtectedContextOnly); // Prose CR needed
    
    //@siclog "Steps 16 - 20" siclog@
    //Check: Does the UE send a correctly composed INVITE request?(Step 1 of annex A.6)
    
    v_IMS_DATA_REQ := f_IMS_EmergencyCallSetup_NR5GC_A_6_Step1(A_2_1_A6,  v_GeolocationUsage);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Steps 16");

    // Emergency call setup: 34.229-5 Annex A.6
    f_IMS_EmergencyCallSetup_NR5GC_A_6_Step2ToEnd(v_IMS_DATA_REQ,  v_GeolocationUsage);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 20");
    

    //@siclog "Steps 21-22" siclog@
    //SS sends BYE.(Step 1 of annex A.8)
    //UE sends 200 OK.(Step 1 of annex A.8)
    v_ContactUrlUE := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);
    f_IMS_CallReleaseMT(v_ContactUrlUE);
    
    f_IMS_TestBody_Set(false);
    
    f_IMS_IMS_SendCoOrdMsg(OtherPDN);
  }




After Change
	       function f_TC_10_9_NR5GC_IMS2() runs on IMS_PTC
  { /* IMS implementation for PDN2 - Emergency call */
    var SipUrl v_ContactUrlUE;
    var template (value) MessageHeader v_MessageHeader;
    var template (value) MessageBody v_MessageBody;
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var REGISTER_Request v_RegisterRequest;
    var boolean v_CondA7 := true;
    var boolean v_GeolocationUsage := pc_IMS_Geolocation; /* the test case intends to use GeoLocationInformation if supported by the UE (PICS) @sic R5s150846 sic@
                                                             INVITE (step 1 of Annex C.22) refers to condition A8 of A.2.1 */

    f_IMS_PTC_Init(PDN_2);
    
    // Test Case Body
    f_IMS_TestBody_Set(true);
    
    f_IMS_IMS_WaitForTrigger(OtherPDN);
    // @siclog "Steps 2-9" siclog@
    // Steps 1-8 of Table 4.9.11.2.2-1 of TS 38.508-1 are performed.
    // EXCEPTION: In parallel to the events described in steps 9-10 below the events specified in step of Table 4.9.11.2.2-2 of TS 38.508-1 [21] take place
    // @siclog "Step 11-12" siclog @
    // Steps 9-10 of Table 4.9.11.2.2-1 of TS 38.508-1 are performed.
    //EXCEPTION: In parallel to the events described in steps 12-20  below the events specified in 11-13 of annex Table 4.9.11.2.2-1 of TS 38.508-1[21] are performed.
    f_IMS_CC_StartCall(IPCAN_EmergencyCall_NoRegistration); // 4.9.11.2.2-2 of TS 38.508-1
    
    //@siclog "Step 12" siclog@
    //UE sends REGISTER(Step 1 of Annex A.3)
    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest(-, omit, -, -, -, v_CondA7);
    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;
    
    // Reconfig of UE client ports in case the UE uses UDP and different than default (5060) ports
    f_IMS_Register_SecurityInit(v_RegisterRequest);
    
    //@siclog "Step 13" siclog@
    //SS sends 401 Unauthorized(Step 2 of Annex A.3)
    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterRequest))));

    //@siclog "Step 14" siclog@
    //UE sends REGISTER(Step 3 of Annex A.3)
    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest(-, -, v_CondA7);
    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;
    
    //@siclog "Step 15" siclog@
    //SS sends 403 Forbidden
    v_MessageHeader := f_IMS_RegisterResponse_403_MessageHeaderTX(v_RegisterRequest);
    v_MessageHeader.contentType := cs_ContentType(c_ims3gppApplication);
    v_MessageBody := cs_MessageBody_XML_Ims_3gpp(cs_Ims_3gpp_Emergency("anonymous-emergencycall"));
    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine403, v_MessageHeader, v_MessageBody)));
    
    f_IMS_GlobalRelease (oldestProtectedContextOnly); // Prose CR needed
    
    //@siclog "Steps 16 - 20" siclog@
    //Check: Does the UE send a correctly composed INVITE request?(Step 1 of annex A.6)
    
    v_IMS_DATA_REQ := f_IMS_EmergencyCallSetup_NR5GC_A_6_Step1(A_2_1_A6,  v_GeolocationUsage);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Steps 16");

    // Emergency call setup: 34.229-5 Annex A.6
    f_IMS_EmergencyCallSetup_NR5GC_A_6_Step2ToEnd(v_IMS_DATA_REQ,  v_GeolocationUsage);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 20");
    
    f_Delay(5.0);
    //@siclog "Steps 21-22" siclog@
    //SS sends BYE.(Step 1 of annex A.8)
    //UE sends 200 OK.(Step 1 of annex A.8)
    v_ContactUrlUE := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);
    f_IMS_CallReleaseMT(v_ContactUrlUE);
    
    f_IMS_CC_EndCall(IPCAN_EmergencyCall_NormalService, IPCAN_SpeechAndEmergencyCall); //signal to IPCAN that call is released and wait for coordination from there

    f_IMS_TestBody_Set(false);
    
    // Inform NR that PDU session can be released (caught by f_NR5GC_EmergencyCallRelease)
    f_IMS_IPCAN_SendCoOrdMsg(IPCAN);
    
    f_IMS_IMS_SendCoOrdMsg(OtherPDN);
  } 




1. [bookmark: _Toc100832114]Change 2
	Function name
	f_TC_10_9_NR5GC_IMS1

	Reason for change
	The f_IMS_CC_Postamble (if called with TestProcedure parameter) is informing IPCAN to release PDU session, however that was already done from IMS2 and doing it again from IMS1 would lead to runtime error as there are no more PDU sessions.

	Summary of change
	f_IMS_CC_Postamble is not called with the TestProcedure parameter.

	TTCN module
	IMS_NR5GC_EmergencyCallTestcases.ttcn

	MCC160 Comment
	Alternative solution implemented: moved existing function call from IMS1 to IMS2



Before Change
	    function f_TC_10_9_NR5GC_IMS1() runs on IMS_PTC
  { // Emergency call without emergency registration / UE credentials are not accepted / 5GS

    var boolean v_GeolocationUsage := pc_IMS_Geolocation;
    
    f_IMS_CC_Preamble(IPCAN_EmergencyCall, IMS_REGISTERED);
    
    // Trigger SS to execute location related functions to allow UE to get its position
    if  (v_GeolocationUsage) {
      f_IMS_IPCAN_UpdateUELocationInformation(IPCAN);
    }
      
    f_IMS_IMS_SendCoOrdMsg(OtherPDN);
    
    f_IMS_IMS_WaitForTrigger(OtherPDN); // wait for trigger after IMS call release
    
   f_IMS_CC_Postamble(IPCAN_EmergencyCall_NoRegistration);
  }



After Change
	    function f_TC_10_9_NR5GC_IMS1() runs on IMS_PTC
  { // Emergency call without emergency registration / UE credentials are not accepted / 5GS

    var boolean v_GeolocationUsage := pc_IMS_Geolocation;
    
    f_IMS_CC_Preamble(IPCAN_EmergencyCall, IMS_REGISTERED);
    
    // Trigger SS to execute location related functions to allow UE to get its position
    if  (v_GeolocationUsage) {
      f_IMS_IPCAN_UpdateUELocationInformation(IPCAN);
    }
      
    f_IMS_IMS_SendCoOrdMsg(OtherPDN);
    
    f_IMS_IMS_WaitForTrigger(OtherPDN); // wait for trigger after IMS call release
    
   f_IMS_CC_Postamble(); //IPCAN was already informed to end the call from IMS2
  }



MCC160 Implementation
function f_TC_10_9_NR5GC_IMS1() runs on IMS_PTC
  { // Emergency call without emergency registration / UE credentials are not accepted / 5GS

    var boolean v_GeolocationUsage := pc_IMS_Geolocation;
    
    f_IMS_CC_Preamble(IPCAN_EmergencyCall, IMS_REGISTERED);
    
    // Trigger SS to execute location related functions to allow UE to get its position
    if  (v_GeolocationUsage) {
      f_IMS_IPCAN_UpdateUELocationInformation(IPCAN);
    }
      
    f_IMS_IMS_SendCoOrdMsg(OtherPDN);
  }

function f_TC_10_9_NR5GC_IMS2() runs on IMS_PTC
  { /* IMS implementation for PDN2 - Emergency call */
    var SipUrl v_ContactUrlUE;
    var template (value) MessageHeader v_MessageHeader;
    var template (value) MessageBody v_MessageBody;
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var REGISTER_Request v_RegisterRequest;
    var boolean v_CondA7 := true;
    var boolean v_GeolocationUsage := pc_IMS_Geolocation; /* the test case intends to use GeoLocationInformation if supported by the UE (PICS) @sic R5s150846 sic@
                                                             INVITE (step 1 of Annex C.22) refers to condition A8 of A.2.1 */

    f_IMS_PTC_Init(PDN_2);
    
    // Test Case Body
    f_IMS_TestBody_Set(true);
    
    f_IMS_IMS_WaitForTrigger(OtherPDN);
    // @siclog "Steps 2-9" siclog@
    // Steps 1-8 of Table 4.9.11.2.2-1 of TS 38.508-1 are performed.
    // EXCEPTION: In parallel to the events described in steps 9-10 below the events specified in step of Table 4.9.11.2.2-2 of TS 38.508-1 [21] take place
    // @siclog "Step 11-12" siclog @
    // Steps 9-10 of Table 4.9.11.2.2-1 of TS 38.508-1 are performed.
    //EXCEPTION: In parallel to the events described in steps 12-20  below the events specified in 11-13 of annex Table 4.9.11.2.2-1 of TS 38.508-1[21] are performed.
    f_IMS_CC_StartCall(IPCAN_EmergencyCall_NoRegistration); // 4.9.11.2.2-2 of TS 38.508-1
    
    //@siclog "Step 12" siclog@
    //UE sends REGISTER(Step 1 of Annex A.3)
    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest(-, omit, -, -, -, v_CondA7);
    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;
    
    // Reconfig of UE client ports in case the UE uses UDP and different than default (5060) ports
    f_IMS_Register_SecurityInit(v_RegisterRequest);
    
    //@siclog "Step 13" siclog@
    //SS sends 401 Unauthorized(Step 2 of Annex A.3)
    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterRequest))));

    //@siclog "Step 14" siclog@
    //UE sends REGISTER(Step 3 of Annex A.3)
    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest(-, -, v_CondA7);
    v_RegisterRequest := v_IMS_DATA_REQ.Request.Register;
    
    //@siclog "Step 15" siclog@
    //SS sends 403 Forbidden
    v_MessageHeader := f_IMS_RegisterResponse_403_MessageHeaderTX(v_RegisterRequest);
    v_MessageHeader.contentType := cs_ContentType(c_ims3gppApplication);
    v_MessageBody := cs_MessageBody_XML_Ims_3gpp(cs_Ims_3gpp_Emergency("anonymous-emergencycall"));
    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine403, v_MessageHeader, v_MessageBody)));
    
    f_IMS_GlobalRelease (oldestProtectedContextOnly); // Prose CR needed
    
    //@siclog "Steps 16 - 20" siclog@
    //Check: Does the UE send a correctly composed INVITE request?(Step 1 of annex A.6)
    
    v_IMS_DATA_REQ := f_IMS_EmergencyCallSetup_NR5GC_A_6_Step1(A_2_1_A6,  v_GeolocationUsage);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Steps 16");

    // Emergency call setup: 34.229-5 Annex A.6
    f_IMS_EmergencyCallSetup_NR5GC_A_6_Step2ToEnd(v_IMS_DATA_REQ,  v_GeolocationUsage);
    f_IMS_PreliminaryPass(__FILE__, __LINE__, "Step 20");
    
    //@siclog "Steps 21-22" siclog@
    //SS sends BYE.(Step 1 of annex A.8)
    //UE sends 200 OK.(Step 1 of annex A.8)
    v_ContactUrlUE := f_IMS_PTC_ImsInfo_GetContactUrl(dialog);
    f_IMS_CallReleaseMT(v_ContactUrlUE);
    
    f_IMS_TestBody_Set(false);
    
    f_IMS_CC_Postamble(IPCAN_EmergencyCall_NoRegistration); //@sic R5s220609 sic@
  }
    

1. [bookmark: _Toc100832115]Change 3
	Function name
	fl_IP_IMS_Server_SendDL

	Reason for change
	In scenarios where IMS security establishment failed but SIP message needs to be sent in DL nonetheless (emergency call with failed/rejected 401 procedure, e.g. tests 10.9 and 10.10) it is needed to assign UE/NW addresses for routing. 

	Summary of change
	Added routing addresses into fl_IP_IMS_Server_SendDL also when only unprotected context is present.

	TTCN module
	IP_PTC\IP_PTC_IMS_Handler.ttcn

	MCC160 Comment
	Alternative solution implemented
The proposed change is just a workaround which would not work when the UE is not using TCP at the unprotected port (in which case the unprotected security context has no information about a TCP connection and therefore there are no IP addresses which could be used).
Instead the RegistrationAddress shall be configured when receiving the INVITE for the emergency call without registration. To avoid, that the RegistrationAddress is set to omit again, there shall be not release of contexts for 'PortsAndSecurityConfig' in a_IP_IMS_Config anymore which seems even to be wrong as in case of GIBA there would be no RegistrationAddress to be used for requests sent in DL.




Before Change
	  
    function fl_IP_IMS_Server_SendDL(inout IMS_Server_Type p_ImsServer,
                                   IMS_RoutingInfo_Type p_RoutingInfo,
                                   IMS_RequestResponse_Type p_RequestResponse,
                                   bitstring p_Data) runs on IP_PTC
  {

…

    if (isvalue(p_RoutingInfo.NW_Address) and isvalue(p_RoutingInfo.UE_Address)) {
      v_NW_Address := p_RoutingInfo.NW_Address;
      v_UE_Address := p_RoutingInfo.UE_Address;
    } else if (ispresent(p_ImsServer.RegistrationAddress)) {              /* @sic R5s160735: fl_ImsServer_UeIsRegistered removed sic@ */
      v_NW_Address := p_ImsServer.RegistrationAddress.NW;
      v_UE_Address := p_ImsServer.RegistrationAddress.UE;
    } else {
      FatalError(__FILE__, __LINE__, "missing IP addresses");
    }
…



After Change
	        function fl_IP_IMS_Server_SendDL(inout IMS_Server_Type p_ImsServer,
                                   IMS_RoutingInfo_Type p_RoutingInfo,
                                   IMS_RequestResponse_Type p_RequestResponse,
                                   bitstring p_Data) runs on IP_PTC
  {

…

    if (isvalue(p_RoutingInfo.NW_Address) and isvalue(p_RoutingInfo.UE_Address)) {
      v_NW_Address := p_RoutingInfo.NW_Address;
      v_UE_Address := p_RoutingInfo.UE_Address;
    } else if (ispresent(p_ImsServer.RegistrationAddress)) {              /* @sic R5s160735: fl_ImsServer_UeIsRegistered removed sic@ */
      v_NW_Address := p_ImsServer.RegistrationAddress.NW;
      v_UE_Address := p_ImsServer.RegistrationAddress.UE;
    } else if (ispresent(p_ImsServer.UnprotectedContext.TcpServer)) {              
      v_NW_Address := p_ImsServer.UnprotectedContext.TcpServer.Local.IpAddr;
      v_UE_Address := p_ImsServer.UnprotectedContext.TcpServer.Remote.IpAddr;
    } else {
      FatalError(__FILE__, __LINE__, "missing IP addresses");
    }
…





MCC160 Implementation

	  function f_IMS_INVITE_ReceiveRequest_Optional(template (present) SipUrl p_ExpectedCalleeUri,
                                                template (present) MessageHeader p_MessageHeader_Invite,
                                                template (present) MessageBody p_MessageBody,
                                                template (omit) float p_WaitDuration := omit,
                                                boolean p_IsReInvite := false,
                                                boolean p_IsEmergencyCallWithoutRegistration := false)
    runs on IMS_PTC return template (omit) IMS_DATA_REQ
  {
    var IMS_DATA_REQ v_IMS_DATA_REQ_ByRef;
    var INVITE_Request v_InviteRequest;
    var SipUrl v_ContactUrl;
    var IP_AddrInfo_Type v_UE_Address;
    var IP_AddrInfo_Type v_NW_Address;
    var boolean v_TimerControl := ispresent(p_WaitDuration);

    timer t_Timer;

    if (v_TimerControl) {
      t_Timer.start(valueof(p_WaitDuration));
    }
    alt {
      [] a_IMS_ReceiveRequest(car_IMS_Invite_Request(cr_INVITE_Request(p_ExpectedCalleeUri, p_MessageHeader_Invite, p_MessageBody)), v_IMS_DATA_REQ_ByRef)
        {
          t_Timer.stop;
          v_InviteRequest := v_IMS_DATA_REQ_ByRef.Request.Invite;
          if (p_IsEmergencyCallWithoutRegistration) {      // emergency call without registration   @sic R5s130756 change 14 sic@
            v_ContactUrl := f_MessageHeader_GetContactSipUrl(v_InviteRequest.msgHeader);   /* @sic R5s150268 change 1 sic@ */
            v_UE_Address := v_IMS_DATA_REQ_ByRef.RoutingInfo.UE_Address;
            v_NW_Address := v_IMS_DATA_REQ_ByRef.RoutingInfo.NW_Address;
            f_IMS_PTC_ImsInfo_Init(v_UE_Address, v_NW_Address);

            f_IMS_PortsAndSecurityConfig(f_SIP_ContactUrl_GetPort(v_ContactUrl), cs_IMS_RegistrationInfo(v_NW_Address, v_UE_Address));          /* @sic R5s150268 change 1 sic@ */
          }
          if (not p_IsReInvite) {      /* @sic R5s140020 change 8.2 sic@ */
            f_IMS_PTC_ImsInfo_DialogInit(dialogRemote, v_InviteRequest.msgHeader);
          } else {
            if (not isvalue(f_IMS_PTC_ImsInfo_CseqGet(dialogRemote))) {  /* @sic R5s140235 sic@ */
              f_IMS_PTC_ImsInfo_CseqSet(dialogRemote, v_InviteRequest.msgHeader.cSeq.seqNumber);
            }
          }
          return v_IMS_DATA_REQ_ByRef;
        }
      [v_TimerControl] t_Timer.timeout {}
    }
    return omit;
  }

	  altstep a_IP_IMS_Config(inout IMS_Server_Type p_ImsServer,
                          IP_IMS_CTRL_PORT p_Port) runs on IP_PTC
  {
    var IMS_CONFIG_REQ v_IMS_CONFIG_REQ;
    var IMS_PortsAndSecurityConfigReq_Type v_PortsAndSecurityConfig;
    var template (omit) IMS_RegistrationInfo_Type v_RegistrationInfo;
    var boolean v_WaitForCNF;

    [] p_Port.receive(car_IMS_InstallKey_REQ) -> value v_IMS_CONFIG_REQ
      {
        p_ImsServer.SecurityKeys := v_IMS_CONFIG_REQ.InstallKey;
        p_Port.send(cas_IMS_InstallKey_CNF);
      }

    [] p_Port.receive(car_IMS_PortsAndSecurityConfig_REQ) -> value v_IMS_CONFIG_REQ
      {
        v_PortsAndSecurityConfig := v_IMS_CONFIG_REQ.PortsAndSecurityConfig;
        v_RegistrationInfo := v_PortsAndSecurityConfig.RegistrationInfo;
        if (not isvalue(p_ImsServer.RegistrationAddress) and isvalue(v_RegistrationInfo)) {  /* @sic R5s150268 change 4 sic@ */
          p_ImsServer.RegistrationAddress.NW := valueof(v_RegistrationInfo.NW_Address);
          p_ImsServer.RegistrationAddress.UE := valueof(v_RegistrationInfo.UE_Address);
        }
        
        if (isvalue(v_PortsAndSecurityConfig.UnprotectedPort_us)) {
          p_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_us := v_PortsAndSecurityConfig.UnprotectedPort_us;   /* set UE's port number as requested by the UE */
        } else {
          p_ImsServer.UnprotectedContext.SecurityInfo.Unprotected.Port_us := tsc_IMS_PortNumber_5060;                       /* restore port number e.g. if it has been set to other value in previous session */
        }
        
        if (isvalue(v_RegistrationInfo.SecurityInfo)) {                                                                     /* non-GIBA   @sic R5s120858 change 6..8 - MCC160 implementation sic@ */
          if (not isvalue(p_ImsServer.SecurityKeys)) {
            FatalError(__FILE__, __LINE__, "security keys missing");
          }
          fl_IMS_Server_AddSecurityContext(p_ImsServer, valueof(v_RegistrationInfo));                                       /* @sic R5-145732: support of 2nd security context sic@ */
        }
}
        /*else {
          fl_IMS_Server_RemoveSecurityContext(p_ImsServer);
        }*/
        p_Port.send(cas_IMS_PortsAndSecurityConfig_CNF);
      }
    [] p_Port.receive(car_IMS_SecurityRelease_REQ) -> value v_IMS_CONFIG_REQ
      {
        v_WaitForCNF := false;
        fl_IMS_Server_RemoveSecurityContexts(p_ImsServer, v_IMS_CONFIG_REQ.SecurityReleaseMode, v_WaitForCNF);              /* @sic R5s170231 - Additional changes sic@ */
        p_Port.send(cas_IMS_SecurityRelease_CNF);
      }
    [] p_Port.receive(car_IMS_CloseTCP_REQ) -> value v_IMS_CONFIG_REQ                                                       /* @sic R5s130988 change 1 - MCC160 implementation sic@ */
      {
        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, unprotected);
        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext1);
        fl_IMS_Server_SecurityContextCloseTCP(p_ImsServer, protectedContext2);
        p_Port.send(cas_IMS_CloseTCP_CNF);
      }
    [] p_Port.receive(car_IMS_RestartPCSCF_REQ) -> value v_IMS_CONFIG_REQ                                                   /* @sic R5-213534: support of 2nd P-CSCF sic@ */
      {
        p_ImsServer := f_IP_IMS_Stop(p_ImsServer);
        p_ImsServer := f_IP_IMS_Start(p_ImsServer,
                                      v_IMS_CONFIG_REQ.RestartPCSCF.IPAddressIPv4,
                                      v_IMS_CONFIG_REQ.RestartPCSCF.IPAddressIPv6,
                                      v_IMS_CONFIG_REQ.RestartPCSCF.PdnIndex);
        p_Port.send(cas_IMS_RestartPCSCF_CNF);
      }
  }



Additional changes by MCC160 for improvement
	Definition
	IMS_SecurityRelease_Type

	Reason for change
	Current use of IMS_SecurityRelease_Type is confusing:
Until now there are only two values
- full: all protected and unprotected security contexts shall be release 
- oldestProtectedContextOnly: only the oldest protected security contexts shall be release 
but there is no value to indicate that the protected context(s) shall be released but the unprotected context shall be kept, which is intended in most cases.
As the majority of test case use only one protected security context, 'oldestProtectedContextOnly' is used for this case so far (what may cause confusion).

	Summary of change
	New value 'keepUnprotected' to clarify the purpose, when all protected contexts shall be released.
As the purpose of f_IMS_GlobalRelease is to release IMS with either releasing the unprotect context too or keeping it, all occurrences of "f_IMS_GlobalRelease(oldestProtectedContextOnly)" are replaced by "f_IMS_GlobalRelease(keepUnprotected)" 
NOTE: in cases where 'oldestProtectedContextOnly' is still needed the function f_IMS_Register_SecurityRelease is used.

	TTCN modules
	IMS_ASP_TypeDefs
IMS_CC_TestcaseImplementation_12
IMS_NR5GC_EmergencyCallTestcases
IMS_NR5GC_CallControlTestcases
IMS_IRAT_19_IMS
IMS_EmergencyCall_IMS



	  type enumerated IMS_SecurityReleaseMode_Type {
    full,
    keepUnprotected,
    oldestProtectedContextOnly
  };

	  function f_IMS_MO_Call_504_ServerTimeout(IMS_SDP_AudioCodecList_Type p_AudioCodecList := tsc_IMS_SDP_AudioCodecList_WBAMR_AMR) runs on IMS_PTC
  {
…
    f_IMS_GlobalRelease(keepUnprotected);
…
  }

	  function f_TC_10_9_NR5GC_IMS2() runs on IMS_PTC
  {
…
    f_IMS_GlobalRelease(keepUnprotected);
…
  }

	  function f_TC_10_10_NR5GC_IMS2() runs on IMS_PTC
  {
…
    f_IMS_GlobalRelease(keepUnprotected);
…
  }

	  function f_TC_7_2_NR5GC_IMS1() runs on IMS_PTC
  {
…
    f_IMS_GlobalRelease(keepUnprotected);
…
  }

	  function f_TC_19_4_5_IMS2() runs on IMS_PTC
  {
…
    f_IMS_GlobalRelease(keepUnprotected);
…
  }

	  function f_TC_11_2_3_IMS1() runs on IMS_PTC
  {
…
    f_IMS_GlobalRelease(keepUnprotected);
…
  }



	Function
	f_IMS_PTC_RegistrationInfo_Reset

	Reason for change
	1. f_IMS_PTC_RegistrationInfo_Reset can be removed as it is redundant:
In all cases where f_IMS_PTC_RegistrationInfo_Reset is called, f_IMS_PTC_Reset is called immediately afterwards resetting vc_IMS_Global.ImsInfo.Registration too.
2. In f_TC_10_10_NR5GC_IMS2 after f_IMS_PTC_Reset is called, f_IMS_GlobalRelease is called which itself calls f_IMS_PTC_Reset 

	Summary of change
	1. f_IMS_PTC_RegistrationInfo_Reset removed from IMS_Component and where being called.
2. f_IMS_PTC_Reset removed from f_TC_10_10_NR5GC_IMS2 

	TTCN modules
	IMS_Component
IMS_CC_EmergencyCallTestcases
IMS_NR5GC_EmergencyCallTestcases



	 /* REMOVED
 function f_IMS_PTC_RegistrationInfo_Reset() runs on IMS_PTC
  {
    vc_IMS_Global.ImsInfo.Registration := valueof(cs_IMS_RegistrationInfo_Init);
  }*/

	  function f_TC_19_4_6_IMS2() runs on IMS_PTC
  {
…  // f_IMS_PTC_RegistrationInfo_Reset();
    f_IMS_PTC_Reset();
…
  }

	  function f_TC_19_4_7_IMS2() runs on IMS_PTC
  {
…
   // f_IMS_PTC_RegistrationInfo_Reset();
    f_IMS_PTC_Reset();
…
  }

	  function f_TC_10_10_NR5GC_IMS2() runs on IMS_PTC
  {
…
   /* f_IMS_PTC_RegistrationInfo_Reset();
    f_IMS_PTC_Reset();*/

    f_IMS_PTC_SetRanType(NR); // Ran Type Condition A28

    f_IMS_GlobalRelease(keepUnprotected);
…
  }




1. [bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665][bookmark: _Toc100832116]Branches executed 
On Anritsu, the test case was executed on NR band n41 using NR ciphering nea1 and integrity algorithm nia1.
3 [bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666][bookmark: _Toc100832117]Execution Log Files 
3.1 [bookmark: _Toc97137458][bookmark: _Toc100832118]Qualcomm SM 8450+SDX65 5G UE
[bookmark: _Hlk90552893]The Qualcomm SDX UE passed this test case on Anritsu Protocol Conformance Test System ME7834NR. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
TC_10_9_LOG.xml
· PICS/PIXIT parameter file:
TC_10_9_PICS.txt


4 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc100832119]References 
	[1]
	R5s220610 - Supporting information for addition of IMS 5GS emergency call  test case 10.9





